Functional programming languages Part III: program transformations

Xavier Leroy

INRIA Paris-Rocquencourt

MPRI 2-4, 2015-2016

Generalities on program transformations

In the broadest sense: all translations between programming languages that preserve the meaning of programs.


Focus in this lecture: examples of program transformations that eliminate high-level features of a (mostly) functional programming language and target a smaller or lower-level language.

I.e. translate from language L_1 having features A, B, C, D to language L_2 having features B, C, D, E.

Uses for program transformations

- As passes in a compiler. Progressively bridge the gap between high-level source languages and machine code.
- To give semantics to the source language. The semantics of feature A is defined in terms of that of features B, C, D, E.
- To program in languages that lack the desired feature A. E.g. use higher-order functions or objects in C; use imperative programming in Haskell or Coq.

Big picture of compilation



Outline

- Closure conversion
- 2 Defunctionalization
- 3 Exception-returning style
- 4 State-passing style
- Continuation-passing style
- 6 Bonus track: CPS + defunctionalization = ?
- Bonus track: callcc + constructive logic = ?

Closure conversion

Goal: make explicit the construction of closures and the accesses to the environment part of closures.

Input: a functional programming language with general functions (possibly having free variables) as first-class values.

Output: the same language where only closed functions (without free variables) are first-class values. Such closed functions can be represented at run-time as code pointers, just as in C for instance.

Idea: every function receives its own closure as an extra argument, from which it recovers values for its free variables. Such functions are closed. Function closures are explicitly represented as a tuple (closed function, values of free variables).

Uses: compilation; functional programming in C, Java, ...

Definition of closure conversion

The translation extends isomorphically to other constructs, e.g.

$$[[let x = a in b]] = let x = [[a]] in [[b]]$$

 $[[a + b]] = [[a]] + [[b]]$

Example of closure conversion

Source program in Caml:

```
fun x lst ->
  let rec map f lst =
    match lst with [] -> [] | hd :: tl -> f hd :: map f tl
  in
    map (fun y -> x + y) lst
```

Result of partial closure conversion for the f argument of map:

Closure conversion for recursive functions

In a recursive function $\mu f.\lambda x.a$, the body a needs access to f, i.e. the closure for itself. This closure can be found in the extra function parameter that closure conversion introduces.

$$\llbracket \mu f. \lambda x.a \rrbracket = \operatorname{tuple}(\lambda f, x. \operatorname{let} x_1 = \operatorname{field}_1(f) \operatorname{in} \\ \ldots \\ \operatorname{let} x_n = \operatorname{field}_n(f) \operatorname{in} \\ \llbracket a \rrbracket, \\ x_1, \ldots, x_n) \\ \text{where } x_1, \ldots, x_n \operatorname{are the free variables of } \mu f. \lambda x.a$$

Note that regular functions $\lambda x.a$ are converted exactly like pseudo-recursive functions $\mu f.\lambda x.a$ where f is a variable not free in a.

Minimal environments in closures

Closures built by closure conversions have minimal environments: they contain values only for variables actually free in the function.

In contrast, closures built by the abstract machines of lecture II have full environments containing values for all variables in scope when the function is evaluated.

Minimal closures consume less memory and enable a garbage collector to reclaim other data structures earlier. Consider:

let 1 =
$$\langle big | list \rangle$$
 in λx . x+1

With full closures, the list 1 is reachable from the closure of $\lambda x.x + 1$ and cannot be reclaimed as long as this closure is live.

With minimal closures, no reference to 1 is kept in the closure, enabling earlier garbage collection.

Closure conversion in object-oriented style

If the target of the conversion is an object-oriented language in the style of Java, we can use the following variant of closure conversion:

Closure conversion in object-oriented style

The class $C_{\lambda x.a}$ (one for each λ -abstraction in the source) is defined as follows:

```
class C_{\lambda_{Y,a}} {
     Object x_1; ...; Object x_n;
     C_{\lambda x,a} (Object x_1, \ldots, Object x_n) {
          this.x_1 = x_1; ...; this.x_n = x_n;
     Object apply(Object x) {
          return [a];
```

Closures and objects

In more general terms:

- ullet Closure pprox Object with a single apply method
- ullet Object pprox Closure with multiple entry points

Both function application and method invocation compile down to self application:

Outline

- Closure conversion
- 2 Defunctionalization
- 3 Exception-returning style
- State-passing style
- 5 Continuation-passing style
- 6 Bonus track: CPS + defunctionalization = ?
- Bonus track: callcc + constructive logic =?

Defunctionalization

Goal: like closure conversion, make explicit the construction of closures and the accesses to the environment part of closures. Unlike closure conversion, do not use closed functions as first-class values.

Input: a functional programming language, with general functions (possibly having free variables) as first-class values.

Output: any first-order language (no functions as values).

Idea: represent each function value $\lambda x.a$ as a data structure $C(v_1, \ldots, v_n)$ where the constructor C uniquely identifies the function, and the constructor arguments v_1, \ldots, v_n are the values of the free variables x_1, \ldots, x_n .

Uses: functional programming in Pascal, Ada, ...

Definition of defunctionalization

(Other constructs: isomorphically.)

Definition of defunctionalization

The apply function collects the bodies of all functions and dispatches on its first argument. There is one case per function occurring in the source program.

```
let rec apply(\mathit{fun}, \mathit{arg}) = match \mathit{fun} with C_{\lambda x.a}(x_1, \ldots, x_n) \to \text{let } x = \mathit{arg} \text{ in } \llbracket a \rrbracket C_{\mu f'.\lambda x'.a'}(x'_1, \ldots, x'_{n'}) \to \text{let } f' = \mathit{fun} \text{ in let } x' = \mathit{arg} \text{ in } \llbracket a' \rrbracket C_{\mu f'.\lambda x'.a'}(x'_1, \ldots, x'_{n'}) \to \text{let } f' = \mathit{fun} \text{ in let } x' = \mathit{arg} \text{ in } \llbracket a' \rrbracket in [\![\mathit{program}]\!]
```

Note: this is a whole-program transformation, unlike closure conversion.

Example

```
Defunctionalization of (\lambda x. \lambda y. x) 1 2:
```

```
let rec apply (fn, arg) =
   match fn with
    | C1() -> let x = arg in C2(x)
    | C2(x) -> let y = arg in x
in
   apply(apply(C1(), 1), 2)
```

We write C1 for $C_{\lambda x, \lambda v, x}$ and C2 for $C_{\lambda v, x}$.

Outline

- Closure conversion
- 2 Defunctionalization
- 3 Exception-returning style
- State-passing style
- 5 Continuation-passing style
- 6 Bonus track: CPS + defunctionalization = ?
- 7 Bonus track: callcc + constructive logic =?

Exceptions

Exceptions are a control structure useful for error reporting, error handling, and more generally all situations that need "early exit" out of a computation.

```
let f x =
  try 1 + (if x = 0 then raise Error else 100 / x)
  with Error -> 101
```

In try a with Error $\rightarrow b$, if a evaluates normally without raising an exception, its value is returned as the value of the try...with. For instance, f 4 = 26.

If a raises the Error exception, control branches to b, which becomes the result of the try...with. For instance, f = 101.

Exceptions

For a more realistic example of early exit, consider the computation of the product of a list of integers, returning 0 as soon as a list element is 0:

```
let product lst =
    let rec prod = function
      | [] -> 1
      | 0 :: tl -> raise Exit
      | hd :: tl -> hd * prod tl
    in
      try prod 1st with Exit -> 0
```

Reduction semantics for exceptions

In Wright-Felleisen style, add two head reduction rules for try...with and a generic exception propagation rule:

$$\begin{array}{ccc} (\text{try } v \text{ with } x \to b) & \stackrel{\varepsilon}{\to} & v \\ (\text{try raise } v \text{ with } x \to b) & \stackrel{\varepsilon}{\to} & b[x \leftarrow v] \\ & F[\text{raise } v] & \stackrel{\varepsilon}{\to} & \text{raise } v & \text{if } F \neq [\,] \end{array}$$

Exception propagation contexts F are like reduction contexts E but do not allow crossing a try...with

Reduction contexts:

$$E ::= [] \mid E \mid b \mid v \mid E \mid \text{try } E \text{ with } x \rightarrow b \mid \dots$$

Exception propagation contexts:

$$F ::= [] | F b | v F | \dots$$

Reduction semantics for exceptions

Assume the current program is $p = E[\mathtt{raise}\ v]$, that is, we are about to raise an exception. If there is a try...with that encloses the raise, the program will be decomposed as

$$p = E'[\text{try } F[\text{raise } v] \text{ with } x \to b]$$

where F contains no try...with constructs.

F[raise v] head-reduces to raise v, and $E'[\text{try } [] \text{ with } x \to b]$ is an evaluation context. The reduction sequence is therefore:

$$p = E'[ext{try } F[ext{raise } v] ext{ with } x o b] o E'[ext{try raise } v ext{ with } x o b] o E'[b[x \leftarrow v]]$$

If there are no try...with around the raise, E is an exception propagation context F and the reduction is therefore

$$p = E[\mathtt{raise}\ v] \to \mathtt{raise}\ v$$

Reduction semantics for exceptions

Considering reduction sequences, a fourth possible outcome of evaluation appears: termination on an uncaught exception.

- Termination: $a \stackrel{*}{\rightarrow} v$
- Uncaught exception: $a \stackrel{*}{\rightarrow} raise v$
- Divergence: $a \stackrel{*}{\rightarrow} a' \rightarrow \dots$
- Error: $a \stackrel{*}{\rightarrow} a' \not\rightarrow$ where $a \neq v$ and $a \neq \texttt{raise } v$.

Natural semantics for exceptions

In natural semantics, the evaluation relation becomes $a \Rightarrow r$ where evaluation results are $r := v \mid \mathtt{raise}\ v$. Add the following rules for $\mathtt{try}...\mathtt{with}$:

$$\cfrac{a\Rightarrow v}{\texttt{try } a \; \texttt{with} \; x \rightarrow b \Rightarrow v} \qquad \cfrac{a\Rightarrow \texttt{raise} \; v' \quad b[x \leftarrow v'] \Rightarrow r}{\texttt{try } a \; \texttt{with} \; x \rightarrow b \Rightarrow r}$$

as well as exception propagation rules such as:

$$\frac{a \Rightarrow \text{raise } v}{a \ b \Rightarrow \text{raise } v} \qquad \qquad \frac{a \Rightarrow v' \quad b \Rightarrow \text{raise } v}{a \ b \Rightarrow \text{raise } v}$$

Conversion to exception-returning style

Goal: get rid of exceptions.

Input: a functional language featuring exceptions (raise and try...with).

Output: a functional language with pattern-matching but no exceptions.

Idea: every expression a evaluates to either

- V(v) if a evaluates normally to a value v
- E(v) if a terminates early by raising exception v.

(V, E are datatype constructors.)

Uses: giving semantics to exceptions; programming with exceptions in Haskell; reasoning about exceptions in Coq.

Definition of the conversion: Core constructs

Effect on types: if $a:\tau$ then $[\![a]\!]:[\![\tau]\!]$ outcome where $[\![\tau_1\to\tau_2]\!]=[\![\tau_1]\!]\to[\![\tau_2]\!]$ outcome and $[\![\tau]\!]=\tau$ for base types and where type 'a outcome = V of 'a | E of exn.

Definition of the conversion: Exception-specific constructs

$$\llbracket \text{raise } a \rrbracket \ = \ \text{match } \llbracket a \rrbracket \text{ with } E(e_a) \to E(e_a) \mid V(v_a) \to E(v_a)$$

$$\llbracket \text{try } a \text{ with } x \to b \rrbracket \ = \ \text{match } \llbracket a \rrbracket \text{ with } E(x) \to \llbracket b \rrbracket \mid V(v_a) \to V(v_a)$$

Example of conversion

```
[try fn arg with exn \rightarrow 0] =
     match
           match V(fn) with
            | E(x) \rightarrow E(x)
            V(x) \rightarrow
                 match V(arg) with
                 | E(y) \rightarrow E(y)
                 V(y) \rightarrow x y
     with
      |V(z)| \rightarrow V(z)
      I E(exn) \rightarrow V(0)
```

Administrative reductions

The naive conversion generates many useless match constructs over arguments whose shape V(...) or E(...) is known at compile-time.

These can be eliminated by performing administrative reductions $\overset{adm}{\rightarrow}$ at compile-time, just after the conversion:

Example of conversion

After application of administrative reductions, we obtain:

```
[try fn arg with exn \rightarrow 0] = match fn arg with | V(z) \rightarrow V(z) | E(exn) \rightarrow V(0)
```

Correctness of the conversion

Define the conversion of a value $\llbracket v \rrbracket_v$ as $\llbracket N \rrbracket_v = N$ and $\llbracket \lambda x.a \rrbracket_v = \lambda x. \llbracket a \rrbracket$

Theorem 1

- ① If $a \Rightarrow v$, then $[a] \Rightarrow V([v]_v)$.
- 2 If $a \Rightarrow \text{raise } v$, then $[a] \Rightarrow E([v]_v)$.
- \bullet If $a \Rightarrow \infty$, then $[a] \Rightarrow \infty$.

Proof.

(1) and (2) are proved simultaneously by induction on the derivation of $a \Rightarrow r$ where r is an evaluation result. (3) is by coinduction. All three proofs use the substitution lemma $[a[x \leftarrow v]] = [a][x \leftarrow [v]_v]$.

Outline

- Closure conversion
- 2 Defunctionalization
- 3 Exception-returning style
- 4 State-passing style
- Continuation-passing style
- 6 Bonus track: CPS + defunctionalization = ?
- 7 Bonus track: callcc + constructive logic =?

State (imperative programming)

The word state in programming language theory refers to the distinguishing feature of imperative programming: the ability to assign (change the value of) variables after their definition, and to modify data structures in place after their construction.

References (recap)

A simple yet adequate way to model state is to introduce references: indirection cells / one-element arrays that can be modified in place. The basic operations over references are:

ref a

Create a new reference containing initially the value of a.

deref a also written la

Return the current contents of reference a.

assign a b also written a := b

Replace the contents of reference a with the value of b. Subsequent deref a operations will return this value.

Uses of references

A let-bound reference emulates an imperative variable:

(We write a; b instead of let z = a in b if z is not free in b.)

Such references also enable a function to maintain an internal state:

```
let make_pseudo_random_generator seed =
  let state = ref seed in
    \lambda n. state := (!state * A + B) mod C; !state mod n
```

Uses of references

Arrays \approx list of references (or better: primitive arrays).

Records with mutable fields \approx tuples of references.

Imperative singly-linked list, with in-place concatenation:

```
type 'a mlist = 'a mlist_content ref
and 'a mlist_content = Nil | Cons of 'a * 'a mlist
let rec concat 11 12 =
  match !11 with
  | Nil \rightarrow 11 := !12
  | Cons(x, r) \rightarrow concat !r 12
```

Uses of references

Memoization = caching of already-computed results.

```
let fib_cache = ref map_empty

let rec fib n =
   match map_find n !fib_cache with
   | Some res -> res
   | None ->
     let res =
        if n < 2 then 1 else fib(n-1) + fib(n-2) in
        fib_cache := map_add n res !fib_cache;
        res</pre>
```

Lazy evaluation = CBV + references

Implementation of lazy evaluation:

```
type 'a state =
       | Unevaluated of unit -> 'a
        | Evaluated of 'a
     type 'a lazy = ('a state) ref
     let force (lz: 'a lazy) : 'a =
       match !lz with
       | Evaluated v -> v
        | Unevaluated f ->
            let v = f() in lz := Evaluated v; v
+ syntactic sugar: lazy e \equiv ref(Unevaluated(fun() \rightarrow e))
```

Semantics of references

Semantics based on substitutions fail to account for sharing between references:

$$\mathtt{let}\ r = \mathtt{ref}\ 1\ \mathtt{in}\ r := 2; !r \ \not\rightarrow \ (\mathtt{ref}\ 1) := 2; !(\mathtt{ref}\ 1)$$

Left: the same reference r is shared between assignment and reading; result is 2.

Right: two distinct references are created, one is assigned, the other read; result is 1.

To account for sharing, we must use an additional level of indirection:

- ref a expressions evaluate to locations ℓ : a new kind of variable identifying references uniquely. (Locations ℓ are values.)
- A global environment called the store associates values to locations.

Reduction semantics for references

Reduction over configurations a / s: pairs of a term a and a store s mapping locations to values.

One-step reduction $a \ / \ s \to a' \ / \ s'$ (read: in initial store s, a reduces to a' and updates the store to s')

Example of reduction sequence

In red: the active redex at every step.

```
let r = \text{ref } 3 in let x = r := !r + 1 in !r / \emptyset

\rightarrow \text{ let } r = \ell \text{ in let } x = r := !r + 1 \text{ in } !r / \{\ell \mapsto 3\}

\rightarrow \text{ let } x = \ell := !\ell + 1 \text{ in } !\ell / \{\ell \mapsto 3\}

\rightarrow \text{ let } x = \ell := 3 + 1 \text{ in } !\ell / \{\ell \mapsto 3\}

\rightarrow \text{ let } x = \ell := 4 \text{ in } !\ell / \{\ell \mapsto 3\}

\rightarrow \text{ let } x = () \text{ in } !\ell / \{\ell \mapsto 4\}

\rightarrow \text{ !} \ell / \{\ell \mapsto 4\}

\rightarrow 4
```

Conversion to state-passing style

Goal: get rid of state.

Input: a functional language with imperative references.

Output: a pure functional language.

Idea: every expression a becomes a function that takes a run-time representation of the current store and returns a pair (result value, updated store).

Uses: give semantics to references; program imperatively in Haskell; reason over imperative code in Cog.

Definition of the conversion: Core constructs

Effect on types: if $a:\tau$ then $\llbracket a \rrbracket$: store $\to \llbracket \tau \rrbracket \times$ store where $\llbracket \tau_1 \to \tau_2 \rrbracket = \llbracket \tau_1 \rrbracket \to$ store $\to \llbracket \tau_2 \rrbracket \times$ store and $\llbracket \tau \rrbracket = \tau$ for base types.

Definition of the conversion:

Constructs specific to references

The operations store_alloc, store_read and store_write provide a concrete implementation of the store.

Example of conversion

Administrative reductions:

$$(\lambda s'.b)s \stackrel{adm}{\to} b[s \leftarrow s'] \text{ if } s \text{ variable}$$

$$(\text{match } (a,s) \text{ with } (x,s') \to b) \stackrel{adm}{\to} \text{let } x = a \text{ in } b[s' \leftarrow s]$$

$$\text{let } x = v \text{ in } b \stackrel{adm}{\to} b[x \leftarrow v]$$

$$\text{let } x = y \text{ in } b \stackrel{adm}{\to} b[x \leftarrow y]$$

Example of translation after administrative reductions:

```
[[ let r = ref 3 in let x = r := !r + 1 in !r ]] = \lambdas. match store_alloc s 3 with (r, s1) -> let t = store_read r s1 in let u = t + 1 in match (\varepsilon, store_write r u s1) with (x, s2) -> (store_read r s2, s2)
```

An implementation of the store

Locations = integers.

Stores = pairs (location, finite map location \mapsto value).

$$\begin{array}{lcl} \texttt{store_alloc} \ v \ (n,m) &=& (n, \ (n+1,m+\{n\mapsto v\})) \\ \\ \texttt{store_read} \ \ell \ (n,m) &=& m(\ell) \\ \\ \texttt{store_write} \ \ell \ v \ (n,m) &=& (n,m+\{\ell\mapsto v\}) \end{array}$$

Typing the store?

Easy: static typing of a monomorphic store, where all values stored in references are of the same type sval:

```
\mathtt{store\_alloc} \quad : \quad \mathtt{sval} \to \mathtt{store} \to \mathtt{location} \times \mathtt{store}
```

 $\mathtt{store_read}$: $\mathtt{location} \to \mathtt{store} \to \mathtt{sval}$

 $\mathtt{store_write} \;\; : \;\; \mathtt{location} \to \mathtt{sval} \to \mathtt{store} \to \mathtt{store}$

Much more challenging (programming exercise III.6): a type-safe implementation of a polymorphic store.

 $\mathtt{store_alloc} \ : \ \forall \alpha. \ \alpha \to \mathtt{store} \to \alpha \ \mathtt{location} \times \mathtt{store}$

 $\mathtt{store_read}$: $\forall \alpha. \ \alpha \ \mathtt{location} \to \mathtt{store} \to \alpha$

 $store_write : \forall \alpha. \ \alpha \ location \rightarrow \alpha \rightarrow store \rightarrow store$

Outline

- Closure conversion
- 2 Defunctionalization
- 3 Exception-returning style
- State-passing style
- Continuation-passing style
- 6 Bonus track: CPS + defunctionalization = ?
- 7 Bonus track: callcc + constructive logic =?

Notion of continuation

Given a program p and a subexpression a of p, the continuation of a is the computations that remain to be done once a is evaluated to obtain the result of p.

It can be viewed as a function: (value of a) \mapsto (value of p).

Example 2

Consider the program p = (1+2) * (3+4).

The continuation of a = (1+2) is $\lambda x. x * (3+4)$.

The continuation of a' = (3+4) is λx . 3*x.

(Not λx . (1+2)*x because 1+2 has already been evaluated to 3.)

Continations and reduction contexts

Continuations closely correspond with reduction contexts in Wright-Felleisen reduction semantics:

If p = E[a] and a can reduce, then the continuation of a is λx . E[x].

Example 3

Consider again p = (1 + 2) * (3 + 4).

$$p = (1+2)*(3+4) = E[1+2]$$
 with $E = []*(3+4)$
 $\rightarrow p' = 3*(3+4) = E'[3+4]$ with $E' = 3*[]$
 $\rightarrow 3*7 \rightarrow 21$

The continuation of 1+2 in p is $\lambda x.E[x] = \lambda x. \ x*(3+4)$. The continuation of 3+4 in p' is $\lambda x.E'[x] = \lambda x. \ 3*x$.

Continuations as first-class values

The Scheme language offers a primitive callcc (call with current continuation) that enables a subexpression a of the program to capture its continuation (as a function 'value of $a' \mapsto$ 'value of the program') and manipulate this continuation as a first-class value.

The expression callcc $(\lambda k.a)$ evaluates as follows:

- The continuation of this expression is passed as argument to $\lambda k.a.$
- Evaluation of a proceeds; its value is the value of callcc $(\lambda k.a)$.
- If, during the evaluation of a or at any later time, we evaluate throw k v, evaluation continues as if callcc $(\lambda k.a)$ returned v. That is, the continuation of the callcc expression is reinstalled and restarted with v as the result provided by this expression.

The types are:

callcc: $\forall \alpha, (\alpha \text{ cont } \rightarrow \alpha) \rightarrow \alpha$ throw: $\forall \alpha \beta, \alpha \text{ cont } \rightarrow \alpha \rightarrow \beta$.

Libraries for lists, sets, and other collection data types often provide an imperative iterator iter, e.g.

```
(* list_iter: ('a -> unit) -> 'a list -> unit *)
let rec list_iter f l =
   match l with
   | [] -> ()
   | head :: tail -> f head; list_iter f tail
```

Using first-class continuations, an existing imperative iterator can be turned into a function that returns the first element of a collection satisfying a given predicate pred.

```
let find pred lst =
    callcc (\lambdak.
         list_iter
              (\lambda x. if pred x then throw k (Some x) else ())
              lst;
         None)
```

If an element x is found such that pred x = true, the throw causes Some x to be returned immediately as the result of find pred lst. If no such element exists, list_iter terminates normally, and None is returned.

The previous example can also be implemented with exceptions. However, callcc adds the ability to backtrack the search.

```
let find pred lst = callcc (\lambdak. list_iter (\lambdax. if pred x then callcc (\lambdak'. throw k (Some(x, k'))) else ()) lst; None)
```

When x is found such that pred x = true, find returns not only x but also a continuation k, which, when thrown, will cause backtracking: the search in 1st restarts at the element following x.

The following use of find will print all list elements satisfying the predicate:

```
let printall pred lst =
  match find pred list with
  | None -> ()
   | Some(x, k) -> print_string x; throw k ()
```

The throw k () restarts find pred list where it left the last time.

First-class continuations

callcc and other control operators are difficult to use directly ("the goto of functional languages"), but in combination with references, can implement a variety of interesting control structures:

- Exceptions (exercise III.7)
- Backtracking.
- Imperative iterators (such as Python's yield).
- Checkpointing and replay debugging.
- Coroutines / cooperative multithreading (next slides).

Coroutines

let task message increment =

Consider a simple subroutine (procedure) that does side-effects:

```
let rec loop n =
    if n \ge 10 then () else begin
      print_string message; print_int n;
      loop(n + increment)
    end
  in loop 0
let _ = task " A" 1; task " B" 2; task " C" 3
Execution is purely sequential, producing
AO A1 A2 A3 A4 A5 A6 A7 A8 A9 B0 B2 B4 B6 B8 C0 C3 C9
```

Coroutines

Coroutines interleave their executions, explicitly yielding control to other coroutines.

```
let task message increment =
  fun () ->
  let rec loop n =
    if n >= 10 then terminate() else begin
      print_string message; print_int n;
      yield();
    loop(n + increment)
    end
  in loop 0

let _ = fork(task " A" 1); fork(task " B" 2); fork(task " C" 3)
```

Execution is interleaved, producing for instance

AO BO A1 CO B2 A2 C3 B4 A3 C6 B6 A4 C9 B8 A5 A6 A7 A8 A9

Implementing coroutines with continuations

Maintain a queue of continuations corresponding to coroutines that have not started yet or have previously yielded control.

```
let queue = ref []
let enqueue x = queue := !queue @ [x]
let dequeue () =
  match !queue with x :: q' -> queue := q'; x | [] -> assert false
let queue_is_empty () =
  match !queue with [] -> true | x :: q' -> false
```

Implementing coroutines with continuations

fork proc executes proc() but enqueues a continuation corresponding to the computation following fork proc.

```
let fork (proc: unit -> unit) =
  callcc (fun k -> enqueue k; proc())
```

yield() yields control within its caller and restarts another suspended computation (if any) or the caller itself (otherwise).

```
let yield () =
  callcc (fun k -> enqueue k; throw (dequeue()) ())
```

terminate() terminates its caller, restarting another suspended computation (if any) or terminating the whole program (otherwise).

```
let terminate () =
  if queue_is_empty()
  then exit 0
  else throw (dequeue()) ()
```

Reduction semantics for continuations

In Wright and Felleisen's style, keep the same head reductions $\stackrel{\varepsilon}{\to}$ and the same context rule as before, and add two whole-program reduction rules (\to) for callca and throw:

$$E[{\tt callcc} \ v] \ o \ E[v \ (\lambda x. E[x])]$$
 $E[{\tt throw} \ k \ v] \ o \ k \ v$

Same evaluation contexts E as before.

Note the non-linear use of the context E:

- The rule for callcc duplicates the current context *E*.
- The rule for throw discards the current context E.

Example of reductions

$$egin{aligned} E[ext{callcc} & (\lambda k. \ 1 + ext{throw} \ k \ 0)] \ &
ightarrow \ E[(\lambda k. \ 1 + ext{throw} \ k \ 0) \ (\lambda x. E[x])] \ &
ightarrow \ E[1 + ext{throw} \ (\lambda x. E[x]) \ 0] \ &
ightarrow \ (\lambda x. E[x]) \ 0 \ &
ightarrow \ E[0] \end{aligned}$$

Note how throw discards the current context $E[1+[\,]]$ and reinstalls the saved context E instead.

Conversion to continuation-passing style (CPS)

Goal: make explicit the handling of continuations.

Input: a call-by-value functional language with callcc.

Output: a call-by-value *or call-by-name*, pure functional language (no callcc).

Idea: every term a becomes a function $\lambda k \dots$ that receives its continuation k as an argument, computes the value v of a, and finishes by applying k to v.

Uses: compilation of callcc; semantics; programming with continuations in Caml, Haskell, ...

CPS conversion: Core constructs

A function $\lambda x.a$ becomes a function of two arguments, x and the continuation k that will receive the value of a.

Effect on types: if $a:\tau$ then $\llbracket a \rrbracket: (\llbracket \tau \rrbracket \to \mathtt{answer}) \to \mathtt{answer}$ where $\llbracket \tau \rrbracket = \tau$ for base types and $\llbracket \tau_1 \to \tau_2 \rrbracket = \llbracket \tau_1 \rrbracket \to (\llbracket \tau_2 \rrbracket \to \mathtt{answer}) \to \mathtt{answer}$.

CPS conversion: Continuation operators

In callcc a, the function value f of a receives the current continuation k both as its argument and as its continuation.

In throw a b, we discard the current continuation k and apply directly the value of a (which is a continuation captured by callcc) to the value of b.

Note (again) the non-linear use of the continuation k: callcc duplicates k, and throw ignores k,

Administrative reductions

The CPS translation $[\![\cdots]\!]$ produces terms that are more verbose than one would naturally write by hand, e.g. in the case of an application of a variable to a variable:

$$[\![f \ x]\!] = \lambda k. \ (\lambda k_1.k_1 \ f) \ (\lambda v_1. \ (\lambda k_2.k_2 \ x) \ (\lambda v_2. \ v_1 \ v_2 \ k))$$

instead of the more natural λk . $f \times k$.

This clutter can be eliminated by performing β reductions at transformation time to eliminate the "administrative redexes" introduced by the translation. In particular, we have

$$(\lambda k. \ k. \ v) \ (\lambda x. \ a) \overset{adm}{\rightarrow} (\lambda x. \ a) \ v \overset{adm}{\rightarrow} a[x \leftarrow v]$$

whenever v is a value or variable.

Examples of CPS translation

```
\llbracket f(f x) \rrbracket
   = \lambda k. f \times (\lambda v. f \vee k))
[\mu fact. \lambda n. if n=0 then 1 else n*fact(n-1)]
   = \lambda k_0 \cdot k_0
        \mu fact. \lambda n. \lambda k. if n = 0 then k 1 else fact (n - 1) (\lambda v. k (n * v)))
[callcc (\lambda k.a)]
   = \lambda k. [a] k
```

Execution of CPS-converted programs

Execution of a program prog is achieved by applying its CPS conversion to the initial continuation $\lambda x.x$:

$$\llbracket prog \rrbracket (\lambda x.x)$$

Theorem 4

If $a \stackrel{*}{\to} v$, then $[\![a]\!] (\lambda x.x) \stackrel{*}{\to} [\![v]\!]_v$. If a diverges, so does $[\![a]\!] (\lambda x.x)$.


Where the CPS translation of values v is:

$$[\![N]\!]_v = N$$
 $[\![\lambda x.a]\!]_v = \lambda x.[\![a]\!]$

Plotkin's proof

G. Plotkin, Call-by-name, call-by-value and the lambda-calculus, TCS 1(2), 1975

A difficult proof based on the following simulation diagram:



a: k, the "colon translation", reduces well-chosen administrative redexes.

A proof using natural semantics

These difficulties with administrative reductions can be avoided by using natural semantics in the premises:

Theorem 5

If $a \Rightarrow v$ and k is a value, then $[a] k \xrightarrow{+} k [v]_v$. If $a \Rightarrow \infty$ and k is a value, then [a] k reduces infinitely.

Proof.

Part 1 is by induction on a derivation of $a\Rightarrow v$. (Exercise.) For part 2, consider the set $X=\{\llbracket a\rrbracket\ k\mid a\Rightarrow \infty \land k \text{ value}\}$ and show that

 $\forall a \in X, \exists a' \in X, a \stackrel{+}{\rightarrow} a'$, using part 1.

The two proofs are suprisingly similar to the proofs of thm 10 and 17 in lecture II (correctness of the Modern SECD using natural semantics). \Box

Danvy and Nielsen's one-pass CPS transformation

Split source terms a into:


```
Atoms: t ::= N \mid x \mid \lambda x.a
```

Serious terms: $s := a_1 a_2$

The one-pass CPS transformation:

Danvy and Nielsen's one-pass CPS transformation

Not only does this CPS transformation produce terms free of administrative redexes, but it also enjoys a simple simulation diagram:



CPS terms

The λ -terms produced by the CPS transformation have a very specific shape, described by the following grammar:

CPS atom: $atom := x \mid N \mid \lambda v. \ body \mid \lambda x. \lambda k. \ body$

CPS body: $body ::= atom \mid atom_1 \mid atom_2 \mid atom_1 \mid atom_2 \mid atom_3 \mid atom_3 \mid atom_2 \mid atom_3 \mid atom_3 \mid atom_3 \mid atom_4 \mid atom_5 \mid atom_5 \mid atom_6 \mid atom_6 \mid atom_6 \mid atom_6 \mid atom_7 \mid atom_7 \mid atom_8 \mid at$

 $\llbracket a \rrbracket$ is an atom, and $\llbracket a \rrbracket$ $(\lambda x.x)$ is a body.

Reduction of CPS terms

CPS atom: $atom := x \mid N \mid \lambda v. \ body \mid \lambda x. \lambda k. \ body$

CPS body: $body ::= atom \mid atom_1 \ atom_2 \mid atom_1 \ atom_2 \ atom_3$

Note that all applications (unary or binary) are in tail-position and at application-time, their arguments are closed atoms, that is, values.

The following reduction rules suffice to evaluate CPS-converted programs:

$$(\lambda x.\lambda k.\ body)\ atom_1\ atom_2 \rightarrow body[x \leftarrow atom_1,\ k \leftarrow atom_2]$$

 $(\lambda v.\ body)\ atom \rightarrow body[v \leftarrow atom]$

These reductions are always applied at the top of the program — there is no need for reduction under a context!

The Indifference theorem

G. Plotkin, Call-by-name, call-by-value and the lambda-calculus, TCS 1(2), 1975

Theorem 6 (Indifference)

A closed CPS-converted program [a] ($\lambda x.x$) evaluates in the same way in call-by-name, in left-to-right call-by-value, and in right-to-left call-by-value.

Proof.

Since closed atoms are values, the reduction rules

$$(\lambda x. \lambda k. \ body) \ atom_1 \ atom_2 \rightarrow body[x \leftarrow atom_1, \ k \leftarrow atom_2]$$

 $(\lambda v. \ body) \ atom \rightarrow body[v \leftarrow atom]$

are admissible both under call-by-value and call-by-name. Since we do not reduce under application nodes, left-to-right or right-to-left evaluation of application makes no difference.

CPS conversion and reduction strategy

CPS conversion encodes the reduction strategy in the structure of the converted terms. For instance, right-to-left call-by-value is obtained by taking

$$\llbracket a \ b \rrbracket = \lambda k. \llbracket b \rrbracket (\lambda v_b. \llbracket a \rrbracket (\lambda v_a. v_a v_b k))$$

and call-by-name is achieved by taking

$$\llbracket x \rrbracket = \lambda k. \ x \ k$$
 (or just $\llbracket x \rrbracket = x$)
 $\llbracket a \ b \rrbracket = \lambda k. \ \llbracket a \rrbracket (\lambda v_a. \ v_a \ \llbracket b \rrbracket \ k)$

Note change of viewpoint:

- in CBV, a variable is a value, so we apply the continuation to this value $(\llbracket x \rrbracket = \lambda k. \ k. x)$
- in CBN, a variable is a suspended computation, so we apply this computation to the continuation ($[\![x]\!] = \lambda k. \times k$)

Compilation of CPS terms

CPS terms can be executed by a stackless abstract machines with components

- a code pointer c
- an environment e
- three registers R_1 , R_2 , R_3 .

Instruction set:

```
ACCESS<sub>i</sub>(n) store n-th field of the environment in R_i

CONST<sub>i</sub>(N) store the integer N in R_i

CLOSURE<sub>i</sub>(c) store closure of c in R_i

TAILAPPLY1 apply closure in R_1 to argument R_2

TAILAPPLY2 apply closure in R_1 to arguments R_2, R_3
```

Compilation of CPS terms

Compilation of atoms $A_i(atom)$ (leaves the value of atom in R_i):

$$egin{array}{lcl} \mathcal{A}_i(\underline{n}) &=& \operatorname{ACCESS}_i(n) \ \mathcal{A}_i(N) &=& \operatorname{CONST}_i(N) \ \mathcal{A}_i(\lambda^1.a) &=& \operatorname{CLOSURE}_i(\mathcal{B}(a)) \ \mathcal{A}_i(\lambda^2.a) &=& \operatorname{CLOSURE}_i(\mathcal{B}(a)) \end{array}$$

Compilation of bodies $\mathcal{B}(body)$:

$$\mathcal{B}(a) = \mathcal{A}_1(a)$$

 $\mathcal{B}(a_1 \ a_2) = \mathcal{A}_1(a_1); \mathcal{A}_2(a_2); \text{TAILAPPLY1}$
 $\mathcal{B}(a_1 \ a_2 \ a_3) = \mathcal{A}_1(a_1); \mathcal{A}_2(a_2); \mathcal{A}_3(a_3); \text{TAILAPPLY2}$

Transitions of the CPS abstract machine

Machine state before					Machine state after				
Code	Env	R_1	R_2	R_3	Code	Env	R_1	R_2	R_3
TAILAPPLY1; c	е	c'[e']	V	1	c'	v.e'	_	_	_
TAILAPPLY2; c	е	c'[e']	v_1	<i>v</i> ₂	c'	$v_2.v_1.e'$	_	_	_
$ACCESS_1(n); c$	е	_	v ₂	<i>V</i> 3	С	е	e(n)	v ₂	<i>V</i> 3
$CONST_1(n); c$	е	_	v ₂	<i>V</i> 3	С	е	N	v ₂	<i>V</i> 3
$CLOSURE_1(c'); c$	е	_	<i>v</i> ₂	<i>V</i> 3	С	е	c'[e]	<i>v</i> ₂	<i>V</i> ₃

(Similarly for the other ACCESS, CONST and CLOSURE instructions.)

Continuations vs. stacks

That CPS terms can be executed without a stack is not surprising, given that the stack of a machine such as the Modern SECD is isomorphic to the current continuation in a CPS-based approach.

$$f x = 1 + g x$$
 $g x = 2 * h x$ $h x = ...$

Consider the execution point where h is entered. In the CPS model, the continuation at this point is

$$k = \lambda v.$$
 k' $(2 * v)$ with $k' = \lambda v.$ k'' $(1 + v)$ and $k'' = \lambda v.v$

In the Modern SECD model, the stack at this point is

$$\underbrace{(\mathtt{MUL};\mathtt{RETURN}).e_g.2}_{\approx k}.\underbrace{(\mathtt{ADD};\mathtt{RETURN}).e_f.1}_{\approx k'}.\underbrace{\varepsilon.\varepsilon}_{\approx k''}$$

Continuations vs. stacks

At the machine level, stacks and continuations are two ways to represent the call chain: the chain of function calls currently active.

- Continuations: as a singly-linked list of heap-allocated closures, each closure representing a function activation. These closures are reclaimed by the garbage collector.
- Stacks: as contiguous blocks in a memory area outside the heap, each block representing a function activation. These blocks are explicitly deallocated by RETURN instructions.

Stacks are more efficient in terms of GC costs and memory locality, but need to be copied in full to implement callcc.

Compiling with continuations, A. Appel, Cambridge University Press, 1992.

Outline

- Closure conversion
- 2 Defunctionalization
- 3 Exception-returning style
- State-passing style
- 5 Continuation-passing style
- 6 Bonus track: CPS + defunctionalization = ?
- \bigcirc Bonus track: callcc + constructive logic = ?

Making functions tail-recursive

Contemporary OS put arbitrary, rather low limitations on stack size.

ightarrow Sometimes, programmers need to rewrite recursive functions so that they are tail recursive and run in constant stack space.

A systematic way to do so:

CPS conversion followed by defunctionalization!

Tail-recursion by hand: list concatenation

Natural definition, runs in stack space O(length(x)):

```
let rec app x y =
  match x with
  | [] -> y
  | x1 :: xs -> x1 :: app xs y
```

Tail-recursive definition, using rev_app (reverse & append):

```
let rec rev_app x y =
  match x with
  | [] -> y
  | x1 :: xs -> rev_app xs (x1 :: y)
let app x y = rev_app (rev_app x []) y
```

Systematic way, 1: CPS transformation

We apply the CPS transformation locally to the app function:

```
let rec cps_app x y k =
  match x with
  | [] -> k y
  | x1 :: xs -> cps_app xs y (fun res -> k (x1 :: res))
```

then pass it the initial continuation:

```
let app x y = cps_app x y (fun res -> res)
```

The result runs in constant stack space but is less readable and less efficient than the rev_app implementation.

Systematic way, 2: defunctionalization

Let's defunctionalize cps_app.

```
type 'a funval =
                             (* fun res -> res *)
  | B of 'a * 'a funval (* fun res -> k (x1 :: res) *)
let rec defun_app x y k =
  match x with
  | [] -> apply k y
  | x1 :: xs \rightarrow defun_app xs y (B(x1, k))
and apply k res =
  match k with
  | A -> res
  \mid B(x1, k') \rightarrow apply k' (x1 :: res)
let app x y = defun_app x y A
```

Systematic way, 3: squinting

Note that the type funval of function "closures" is isomorphic to list.

```
let rec defun_app x y k =
   match x with
   | [] -> apply k y
   | x1 :: xs -> defun_app xs y (x1 :: k)

and apply k res =
   match k with
   | [] -> res
   | x1 :: k' -> apply k' (x1 :: res)

let app x y = defun_app x y []
```

Note that apply is really rev_app, and defun_app x y k is really rev_app (rev_app x k) y ...

Tail-recursive interpreters

(Ager, Biernacki, Danvy and Midtgaard, 2003)

What happens if we apply the "CPS + defunctionalization" trick to an interpreter for a functional language?

Let's see on an interpreter for call-by-name, using environments and de Bruijn indices.

A CBN interpreter with environments

```
type term = Var of int | Lam of term | App of term * term
type thunk = Thunk of term * env
 and env = thunk list
 and value = Clos of term * env
let rec eval (e: env) (a: term) : value =
  match a with
  | Var n ->
      let (Thunk(a', e')) = List.nth e n in eval e' a'
  | I.am a ->
      Clos(a, e)
  | App(b, c) ->
      let (Clos(d, e')) = eval e b in eval (Thunk(c, e) :: e') d
```

After CPS conversion . . .

```
let rec cps_eval e a k =
  match a with
  | Var n ->
     let (Thunk(a', e')) = List.nth e n in cps_eval e' a' k
  | Lam a ->
     k (Clos(a, e))
  | App(b, c) ->
     cps_eval e b (fun Clos(d, e') ->
     cps_eval (Thunk(c, e) :: e') d k)

let eval e a = cps_eval e a (fun res -> res)
```

... then defunctionalization ...

```
type funval =
           (* fun res -> res *)
  I A
  | B of term * env * funval
                  (* fun (Clos...) -> ... c ... e ... k *)
let rec defun eval e a k =
  match a with
  | Var n -> let (Thunk(a', e')) = List.nth e n in defun_eval e' a' k
  | Lam a -> apply k (Clos(a, e))
  | App(b, c) \rightarrow defun_eval e b (B(c, e, k))
and apply k res =
  match k with
  | A -> res
  \mid B(c, e, k) ->
      let (Clos(d, e')) = res in defun_eval (Thunk(c, e) :: e') d k
```

let eval e a = defun eval e a A

(funval isomorphic to thunk list; inlining of apply)

...and a bit of cleanup ...

```
let rec tail_eval e a k =
  match a, k with
  | Var n, _ -> let (Thunk(a', e')) = List.nth e n in tail_eval e' a' k
  | Lam a, [] -> Clos(a, e)
  | Lam a, Thunk(c, e') :: k -> tail_eval (Thunk(c, e') :: e) a k
  | App(b, c), _ -> tail_eval e b (Thunk(c, e) :: k)
```

let eval e a = tail eval e a []

... we obtain Krivine's machine!

As can be seen by extracting from tail_eval the underlying transition function over (e, a, k) triples.

```
type step_result = Finished of value | Next of env * term * thunk list
let step = function
  | (e, Var n, k) -> let (Thunk(a', e')) = List.nth e n in Next(e', a', k)
  (e, Lam a, []) -> Finished(Clos(a, e))
  | (e, Lam a, Thunk(c, e') :: k) -> Next(Thunk(c, e') :: e, a, k)
  | (e, App(b, c), k) \rightarrow Next(e, b, Thunk(c, e) :: k)
let rec tail_eval e a k =
 match step (e, a, k) with
  | Finished v -> v | Next(e',a',k') -> tail_eval e' a' k'
```

(Compare with the ACCESS, GRAB and PUSH transitions of Krivine's machine.)

What about call-by-value?

If we apply the same trick to the CBV interpreter from lecture I, we do not obtain the Modern SECD, but something that is very close to another CBV abstract machine: the CEK machine of Felleisen and Friedman.

A Functional Correspondence between Evaluators and Abstract Machines.

Mads Sig Ager, Dariusz Biernacki, Olivier Danvy and Jan Midtgaard. PPDP 2003.

Outline

- Closure conversion
- 2 Defunctionalization
- 3 Exception-returning style
- State-passing style
- 5 Continuation-passing style
- 6 Bonus track: CPS + defunctionalization = ?
- Bonus track: callcc + constructive logic = ?

Propositions-as-types

(The Curry-Howard isomorphism)

Type expressions look a lot like logical formulas:

Types		Logical propositions
Function type $A \rightarrow B$	\approx	Implication $A o B$
Product type $A \times B$	\approx	Conjunction $A \wedge B$
Sum type $A + B$	\approx	Disjunction $A \lor B$
Polymorphic type $\forall \alpha. A$	\approx	Universal quantification $\forall \alpha. A$

Where product types and sum types can be defined as

Proofs-as-programs, propositions-as-types

Likewise, constructive proofs of propositions look a lot like terminating programs of the corresponding type.

A proof of		is		
$A \rightarrow B$	\approx	a total function from proofs of A to proofs of B .		
$A \wedge B$	\approx	a pair of proofs, one for A and another for B .		
$A \lor B$	\approx	a computation that decides which of A and B hold and returns either Left with a proof of A or Right with a proof of B .		
$\forall x : A. \ B(x)$	\approx	a total function from values $v:A$ to proofs of $B(v)$.		

(Much more details in course 2-7-1, "Foundations of proof systems".)

Curry-Howard isomorphism for classical logic

Is it possible to extend this correspondence to classical (not just constructive) logic?

For example: is there a program that "implements" the law of excluded middle?

$$\forall P. P \lor \neg P$$

Answer: yes, but we need callcc or similar control operators!

A Formulæ-as-Types Notion of Control, T. Griffin, POPL 1990.

"Implementing" excluded middle

Modulo Curry-Howard, the law of excluded middle

$$\forall P. P \lor \neg P \equiv \forall P. P \lor (P \rightarrow False)$$

corresponds to the type

$$\forall P. P + (P \rightarrow \mathtt{False})$$

where + is the sum type (Left and Right constructors) and False an empty type.

The following term "implements" excluded middle:

$$callcc(\lambda k. Right(\lambda p. throw k (Left(p))))$$

"Implementing" excluded middle

$$\operatorname{callcc}(\lambda k. \operatorname{Right}(\lambda p. \operatorname{throw} k (\operatorname{Left}(p))))$$

How does it work?

- The rest of the proof term asks "P or not P?"
- The term above returns $Right(\lambda p...)$, i.e. " $P \to False$ ".
- The only thing the rest of the proof can do with this proof is to apply it to a proof p of P to derive a contradiction.
- ... at which time the continuation is invoked, restarting the original proof with Left(p), i.e. "P is true".